Advanced Engineering Mathematics 1.
2-hours exam December 5, 2016.

JE/JKL 05.12.16

Problem 1

```
> restart;with(LinearAlgebra):
```

An inhomogeneous system of linear equations consisting of three equations in four unknowns x_1, x_2, x_3 og x_4 has the augmented matrix $\mathbf{T} = [\mathbf{A}|\ \mathbf{b}]$ given by

```
> T:=a-><<1,0,0>|<0,a,0>|<1,-2,1-a^2>|<3,-4,0>|<0,2,1-a>>:
>  T(a);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & a & -2 & -4 & 2 \\ 0 & 0 & -a^2+1 & 0 & 1-a \end{bmatrix}$$

where $a \in \mathbb{R}$.

Question 1

For $a = -1$ the augmented matrix is

```
> T(-1);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & -1 & -2 & -4 & 2 \\ 0 & 0 & 0 & 0 & 2 \end{bmatrix}$$

The system of equations has no solutions, since $\rho(\mathbf{A}(-1)) = 2 < 3 = \rho(\mathbf{T}(-1))$.
Or. The equations corresponding to the last row in $\mathbf{T}$ has no solutions. Therefore the system of equations has no solutions.
With Maple

```
> LinearSolve(T(-1));
Error, (in LinearAlgebra:-LinearSolve) inconsistent system
```

For $a = 0$ the augmented matrix is

```
> T(0);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & 0 & -2 & -4 & 2 \\ 0 & 0 & 1 & 0 & 1 \end{bmatrix}$$

that has the totally reduced form

```
> trap('T(0)')=ReducedRowEchelonForm(T(0));
```

$$trap(T(0)) = \begin{bmatrix} 1 & 0 & 0 & 0 & 2 \\ 0 & 0 & 1 & 0 & 1 \\ 0 & 0 & 0 & 1 & -1 \end{bmatrix}$$

Since the coefficient matrix and the augmented matrix both have the rank $\rho = 3$ and since $n - \rho = 4 - 3 = 1$ there are infinitely many solutions characterized by a single free parameter.
Putting $x_2 = t$ it is readily seen that

$(x_1, x_2, x_3, x_4) = (2, 0, 1, -1) + t(0, 1, 0, 0)$, $t \in \mathbb{R}$.

With Maple one gets

```
> x=LinearSolve(T(0),free=t);
```

$$x = \begin{bmatrix} 2 \\ t_2 \\ 1 \\ -1 \end{bmatrix}$$

For $a = 1$ the augmented matrix is

```
> T(1);
```

$$\begin{bmatrix} 1 & 0 & 1 & 3 & 0 \\ 0 & 1 & -2 & -4 & 2 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

that already is in its totally reduced form.

Since the coefficient matrix and the augmented matrix both have the rank $\rho = 2$ and since $n - \rho = 4 - 2 = 2$ there are infinitely many solutions characterized by two free parameters. Putting $x_3 = t_1$ and $x_4 = t_2$ it is readily seen that

$(x_1, x_2, x_3, x_4) = (0, 2, 0, 0) + t_1(-1, 2, 1, 0) + t_2(-3, 4, 0, 1)$, $t_1, t_2 \in \mathbb{R}$.

With Maple one gets

```
> x=LinearSolve(T(1),free=t);
```

$$x = \begin{bmatrix} -t_3 - 3\,t_4 \\ 2 + 2\,t_3 + 4\,t_4 \\ t_3 \\ t_4 \end{bmatrix}$$

Question 2

The coefficient matrix is

```
> A:=SubMatrix(T(a),1..3,1..4);
```

$$A := \begin{bmatrix} 1 & 0 & 1 & 3 \\ 0 & a & -2 & -4 \\ 0 & 0 & -a^2 + 1 & 0 \end{bmatrix}$$

and the right-hand-side is

```
> b:=SubMatrix(T(a),1..3,5..5);
```

$$b := \begin{bmatrix} 0 \\ 2 \\ 1 - a \end{bmatrix}$$

$\mathbf{x}_0 = (x_1, x_2, x_3, x_4) = (1, 0, -1, 0)$ is a solution $\Leftrightarrow$ $\mathbf{A}\mathbf{x}_0 = \mathbf{b}$.

```
> x0:=<1,0,-1,0>;
```

$$x0 := \begin{bmatrix} 1 \\ 0 \\ -1 \\ 0 \end{bmatrix}$$

```
> A.x0=b;
```

$$\begin{bmatrix} 0 \\ 2 \\ a^2 - 1 \end{bmatrix} = \begin{bmatrix} 0 \\ 2 \\ 1 - a \end{bmatrix}$$

From this it is seen that $\mathbf{Ax}_0 = \mathbf{b} \Leftrightarrow a^2 - 1 = 1 - a \Leftrightarrow a^2 + a - 2 = 0 \Leftrightarrow a = 1$ or $a = -2$.
So $(x_1, x_2, x_3, x_4) = (1, 0, -1, 0)$ is a solution to the system of equations $\Leftrightarrow a = 1$ or $a = -2$.

Problem 2

```
> restart;with(LinearAlgebra):
```

In $\mathbb{R}^3$ the vectors $\mathbf{v}1 = (-1, 1, 0)$, $\mathbf{v}2 = (1, -2, 1)$, $\mathbf{v}3 = (1, -4, 0)$ and $\mathbf{u} = 2\,\mathbf{v}1 - \mathbf{v}2$ are given.

Question 1

The coordinate matrix for the vectors **v**1, **v**2 og **v**3 with respect to the standardbasis *e* is

```
> eV:=<<-1,1,0>|<1,-2,1>|<1,-4,0>>;
```

$$eV := \begin{bmatrix} -1 & 1 & 1 \\ 1 & -2 & -4 \\ 0 & 1 & 0 \end{bmatrix}$$

Since

```
> rho('eV')=Rank(eV);
```

$$\rho(eV) = 3$$

v1, **v**2 og **v**3 are three linearly independent vectors in $\mathbb{R}^3$ and thus v = (**v**1, **v**2, **v**3) is a basis for $\mathbb{R}^3$.
The change of basis matrix that changes from *v*-coordinates to standard *e*-coordinates is

```
> eMv:=eV;
```

$$eMv := \begin{bmatrix} -1 & 1 & 1 \\ 1 & -2 & -4 \\ 0 & 1 & 0 \end{bmatrix}$$

and the change of basis matrix that changes from standard *e*-coordinates to *v*-coordinates are

```
> vMe:=eMv^(-1);
```

$$vMe := \begin{bmatrix} -\frac{4}{3} & -\frac{1}{3} & \frac{2}{3} \\ 0 & 0 & 1 \\ -\frac{1}{3} & -\frac{1}{3} & -\frac{1}{3} \end{bmatrix}$$

Question 2

The coordinate vector for **u** = 2**v**1 - **v**2 with respect to basis v is

```
> vu:=<2,-1,0>;
```

$$vu := \begin{bmatrix} 2 \\ -1 \\ 0 \end{bmatrix}$$

The coordinate vector for **u** with respect tot the standard basis e is

```
> eu:=eMv.vu;
```

$$eu := \begin{bmatrix} -3 \\ 4 \\ -1 \end{bmatrix}$$

This could also have been achieved by

```
> eu:=2*<-1,1,0>-<1,-2,1>;
```

$$eu := \begin{bmatrix} -3 \\ 4 \\ -1 \end{bmatrix}$$

$f: \mathbb{R}^3 \to \mathbb{R}^3$ is linear and it applies that $f(\mathbf{v}1) = 3\mathbf{v}1$, $f(\mathbf{v}2) = 3\mathbf{v}2$ and $f(\mathbf{v}3) = \mathbf{0}$.

Question 3

Since the three vectors are proper vectors, it follows that **v**1 and **v**2 are eigenvectors for f corresponding to the eigenvalue 3 and that **v**3 is an eigenvector for f corresponding to the eigenvalue 0.
Thus $v = (\mathbf{v}1, \mathbf{v}2, \mathbf{v}3)$ is a basis for $\mathbb{R}^3$ consisting of eigenvectors for f.

Question 4

Since $f(\mathbf{v}1) = 3\mathbf{v}1 = 3\ \mathbf{v}1 + 0\ \mathbf{v}2 + 0\ \mathbf{v}3$, $f(\mathbf{v}2) = 3\mathbf{v}2 = 0\ \mathbf{v}1 + 3\ \mathbf{v}2 + 0\ \mathbf{v}3$ and $f(\mathbf{v}3) = \mathbf{0} = 0\ \mathbf{v}1 + 0\ \mathbf{v}2 + 0\ \mathbf{v}3$ then

```
> vFv:=DiagonalMatrix(<3,3,0>);
```

$$vFv := \begin{bmatrix} 3 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> eFe:=eMv.vFv.vMe;
```

$$eFe := \begin{bmatrix} 4 & 1 & 1 \\ -4 & -1 & -4 \\ 0 & 0 & 3 \end{bmatrix}$$

Question 5

Since both **v**1 and **v**2 belongs to the eigenvectorspace E_3 (they span E_3), then also the linear combination $\mathbf{u} = 2\ \mathbf{v}1 - \mathbf{v}2$ belongs to E_3 .
Therefore **u** is an eigenvector for f corresponding to the eigenvalue 3.

Check

```
> eFe.eu=3*eu;
```

$$\begin{bmatrix} -9 \\ 12 \\ -3 \end{bmatrix} = \begin{bmatrix} -9 \\ 12 \\ -3 \end{bmatrix}$$

Exercise 3

```
> restart;with(LinearAlgebra):
```

$f: C^{\infty}(\mathbb{R}) \to C^{\infty}(\mathbb{R})$ linear and given by

```
> f:=x->diff(x,t)-t*x:
> 'f(x(t))'=f(x(t));
```

$$f(x(t)) = \frac{d}{dt} x(t) - t\, x(t)$$

Question 1

```
> f(x(t))=0;
```

$$\frac{d}{dt} x(t) - t\, x(t) = 0$$

Since

```
> Int(-t,t)=int(-t,t);
```

$$\int (-t)\, dt = -\frac{1}{2} t^2$$

all solutions are

```
> x(t)=c*exp(1/2*t^2);
```

$$x(t) = c\, e^{\frac{1}{2} t^2}$$

where $t \in \mathbb{R}$ and $c \in \mathbb{R}$.
Direct with Maple one gets

```
> dsolve(f(x(t))=0,x(t));
```

$$x(t) = \_C1\, e^{\frac{1}{2} t^2}$$

$P_2(\mathbb{R})$ with the monomial basis $m = (1, t, t^2)$ and $P_3(\mathbb{R})$ with the monomial basis $m = (1, t, t^2, t^3)$.
We now restrict f to be a map of $P_2(\mathbb{R})$ into $C^{\infty}(\mathbb{R})$.

Question 2

The images of the basis vectors are

```
> f(1);
```

$$-t$$

```
> f(t);
```

$$-t^2 + 1$$

```
> f(t^2);
```

$$-t^3 + 2\,t$$

Since $f(1)$, $f(t)$ and $f(t^2)$ all belong to $P_3(\mathbb{R})$, then the image space $f(P_2(\mathbb{R}))$ - that is spanned by these three images - ia a subspace of $P_3(\mathbb{R})$.

Inspired by question 2, we finally let f be a map of $P_2(\mathbb{R})$ into $P_3(\mathbb{R})$.

Question 3

Since $f(1) = -t$, $f(t) = 1 - t^2$ and $f(t^2) = 2t - t^3$ according to question 2, then we see that

```
> mFm:=<<0,-1,0,0>|<1,0,-1,0>|<0,2,0,-1>>;
```

$$mFm := \begin{bmatrix} 0 & 1 & 0 \\ -1 & 0 & 2 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

$P(t) \in \ker(f) \Leftrightarrow f(P(t)) = 0\mathit{1} + 0t + 0t^2 + 0t^3 \Leftrightarrow \text{mFm mP}(t) = \mathbf{0} \in \mathbb{R}^4$.

```
> LinearSolve(<mFm|<0,0,0,0>>);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

I.e. $\ker(f)$ only consists of the zero-polynomial $0\mathit{1} + 0\,t + 0\,t^2$ in $P_2(\mathbb{R})$.

From the dimensional theorem it then follows that $\dim(f(P_2(\mathbb{R}))) = \dim(P_2(\mathbb{R})) - \dim(\ker(f)) = 3 - 0 = 3$.

(Or $\dim(f(P_2(\mathbb{R}))) = \rho(\text{m}\mathbf{F}\text{m}) = 3$)

Comparison with exercise 1: The difference follows from the fact that the only function of the form $c \exp(\frac{1}{2}\,t^2)$, that belongs to our domain $P_2(\mathbb{R})$, is the one that corresponds to $c = 0$.

Problem 4

```
> restart;with(LinearAlgebra):with(plots):
```

$$\begin{bmatrix} x_1'(t) \\ x_2'(t) \end{bmatrix} = \mathbf{A} \begin{bmatrix} x_1(t) \\ x_2(t) \end{bmatrix}, \ t \in \mathbb{R}.$$

where $\mathbf{A}$ is a real 2×2-matrix. It is further given that $\mathbf{A}$ has the eigenvector

```
> v1:=<1+I,2>;
```

$$v1 := \begin{bmatrix} 1 + \mathrm{I} \\ 2 \end{bmatrix}$$

corresponding to the eigenvalue

```
> lambda1:=-1+2*I;
```

$$\lambda 1 := -1 + 2\,\mathrm{I}$$

Question 1

Since **A** is a real 2×2-matrix then the other eigenvalue for **A** is the conjugate of λ1

```
> lambda2:=conjugate(lambda1);
```

$$\lambda 2 := -1 - 2\,\mathrm{I}$$

and a corresponding eigenvector for **A** is the conjugate of **v**1

```
> v2:=evalc(conjugate(v1));
```

$$v2 := \begin{bmatrix} 1-\mathrm{I} \\ 2 \end{bmatrix}$$

Since **v**1 and **v**2 are two linearly independent eigenvectors for **A** the complete complex solution the system of differential equations is

$\mathbf{x}(t) = c1\; e^{(-1+2\,i)t}\,\mathbf{v}1 + c2\; e^{(-1-2\,i)t}\,\mathbf{v}2 \;\;, t \in \mathbb{R} \;\;, c1, c2 \in \mathbb{C}.$

Written in Maple

```
> x:=unapply(c1*exp(lambda1*t)*v1+c2*exp(lambda2*t)*v2,t):
> 'x(t)'=x(t);
```

$$x(t) = \begin{bmatrix} (1+\mathrm{I})\,c1\,\mathrm{e}^{(-1+2\,\mathrm{I})\,t} + (1-\mathrm{I})\,c2\,\mathrm{e}^{(-1-2\,\mathrm{I})\,t} \\ 2\,c1\,\mathrm{e}^{(-1+2\,\mathrm{I})\,t} + 2\,c2\,\mathrm{e}^{(-1-2\,\mathrm{I})\,t} \end{bmatrix}$$

Note that $\mathbf{x}(t)$ is real, if and only if it is equal to its conjugate which is the case if and only if $c1$ and $c2$ are mutually conjugates.

Question 2

From the figure it is seen that $\mathbf{x}(0) = (x_1(0), x_2(0)) = (-1, 4)$.

For the determination of the constants $c1$ and $c2$ we have the linear system of equations

```
> x(0)=<-1,4>;
```

$$\begin{bmatrix} (1+\mathrm{I})\,c1 + (1-\mathrm{I})\,c2 \\ 2\,c1 + 2\,c2 \end{bmatrix} = \begin{bmatrix} -1 \\ 4 \end{bmatrix}$$

with the correpsonding augmented matrix

```
> T:=<<1+I,2>|<1-I,2>|<-1,4>>;
```

$$T := \begin{bmatrix} 1+\mathrm{I} & 1-\mathrm{I} & -1 \\ 2 & 2 & 4 \end{bmatrix}$$

that has the totally reduced form

```
> trap('T'):=ReducedRowEchelonForm(T);
```

$$trap(T) := \begin{bmatrix} 1 & 0 & 1+\frac{3}{2}\,\mathrm{I} \\ 0 & 1 & 1-\frac{3}{2}\,\mathrm{I} \end{bmatrix}$$

From this we see that

```
> c1:=1+3/2*I;
```

$$c1 := 1 + \frac{3}{2}\,\mathrm{I}$$

and

```
> c2:=1-3/2*I;
```

$$c2 := 1 - \frac{3}{2}\,\mathrm{I}$$

c1 and *c2* are mutually conjugates as expected. Cf. the remark in Question1.
The solution which for is then

```
> 'x(t)'=simplify(evalc(x(t)));
```

$$x(t) = \begin{bmatrix} -\mathrm{e}^{-t}\,(\cos(2\,t) + 5\,\sin(2\,t)) \\ 2\,\mathrm{e}^{-t}\,(2\,\cos(2\,t) - 3\,\sin(2\,t)) \end{bmatrix}$$

or written out

```
> 'x[1](t)'=simplify(evalc(x(t)[1]));
```

$$x_1(t) = -\mathrm{e}^{-t}\,(\cos(2\,t) + 5\,\sin(2\,t))$$

```
> 'x[2](t)'=simplify(evalc(x(t)[2]));
```

$$x_2(t) = 2\,\mathrm{e}^{-t}\,(2\,\cos(2\,t) - 3\,\sin(2\,t))$$

where $t \in \mathbb{R}$.

```
> p1:=plot(x(t)[1],t=0..6,color=red):
> p2:=plot(x(t)[2],t=0..6,color=blue):
> display(p1,p2,scaling=constrained,view=-3..5);
```